File And Find

```
199 REM++++++++++CBM C64++++++++++
200 REM+       WRITE      FILES      +
201 REM++++++++++CBM C64++++++++++
220 GOSUB 1500
240 FOR K=65 TO 90
260 Z$=CHR$(K)+X$:C$=D$+"WRITING "+CHR$(K)
280 GOSUB 2000
300 NEXT K
399 REM++++++++++++++++++++++++++++
400 REM+        READ     FILES       +
401 REM++++++++++++++++++++++++++++
420 FOR L=0 TO 1 STEP 0:FOR M=1 TO 1
440 PRINT D$;"ACCESS TO RECORDS"
460 INPUT"SEARCH STRING ( *=QUIT )";N$
480 L$=LEFT$(N$,1):IF L$="*" THEN GOTO 5000
500 IF L$<"A" OR L$>"Z" THEN M=0
520 NEXT M
540 Z$=L$+Y$
560 GOSUB 3000
580 PRINT TAB(5)"RECORD ";N;"(HIT ANY KEY)"
600 GET GT$:IF GT$="" THEN 600
620 NEXT L
999 END
1499 REM*****************************
1500 REM******INITIALISE  S/R********
1501 REM*****************************
1520 D$=CHR$(147):PRINT D$,CHR$(8);CHR$(142)
1540 X$=",S,W":Y$=",S,R"
1600 RETURN
1999 REM*****************************
2000 REM******WRITE A FILE S/R********
2001 REM*****************************
2020 PRINT C$:INPUT"HOW MANY RECORDS";R
2040 OPEN 8,8,2,Z$
2060 IF R=0 THEN PRINT#8,"*":CLOSE8:RETURN
2080 FOR I=1 TO R
2100 PRINT C$:PRINT "RECORD #";I
2120 INPUT "TEXT.....";R$
2140 PRINT#8,R$
2160 NEXT I
2180 PRINT#8,"*":CLOSE8
2499 RETURN
2999 REM*****************************
3000 REM******READ  A FILE S/R********
3001 REM*****************************
3020 PRINT D$;"SEARCHING ";L$;" FOR ";N$
3040 OPEN 8,8,2,Z$
3060 FOR I=1 TO 100000
3080 INPUT#8,R$
3100 PRINT R$
3120 IF R$="*" THEN N=0:I=100000
3140 IF R$=N$ THEN N=1:I=100000
3160 NEXT I:CLOSE8:N$=""
3180 RETURN
5000 REM******CLOSE     PROGRAM********
5020 PRINT CHR$(9);"END OF PROGRAM":STOP
```

220: Initialise

260-280: Create files "A" to "Z"

420-540: Input search record, find initial letter, identify file containing it

560: Search that file

580-600: Report outcome of search

1520: Clear screen, set upper case mode

2040: OPEN file to WRITE

2060: Check for empty file, write "*", CLOSE file

2120-2140: Input text of record, write it to file

2180: Write "*" as last record, CLOSE file

3040: OPEN file to READ

3120: Test for last record, quit search

3140: Test whether record found, quit search

3160: CLOSE file

This program demonstrates the use of sequential files on disk by creating a simple alphabetical index book comprising 26 files, one for each letter of the alphabet. Into each file you can type records beginning with that letter, or no records at all. You can then search for any record. The appropriate file will be searched and displayed until the record is found; if it is not found, the message 'RECORD 0' will be displayed.

The program uses the disk operating system to gain direct access to the file appropriate to the search record; the file itself is then read sequentially, however. This cuts search time to a reasonable length; if the files were on tape, searching would be unbearably slow



ROY INGRAMS

Basic Flavours

BBC Micro

Follow the Spectrum variations for lines 260 and 540
Replace PRINT#8, INPUT#8, and CLOSE8 by PRINT#C8, INPUT#C8, and CLOSE#C8

```
600 GT$=GET$
1520 *DISK
1530 MODE 7
1540 D$=CHR$(12):PRINT D$"USE UPPER CASE"
1550 PRINT"--HIT ANY KEY--":GT$=GET$
2040 C8=OPENOUT(Z$)
3040 C8=OPENIN(Z$)
5020 PRINT"END OF PROGRAM"
```

Spectrum Microdrive

Insert LET where necessary.
Replace PRINT D$;.. by CLS PRINT..
Replace PRINT C$ by CLS:PRINT: C$
Replace OPEN 8,8,2,Z$ by OPEN #8;"m";1;Z$
Replace CLOSE8 by CLOSE #8
Delete line 1540

```
260 Z$=CHR$(K):LET C$="WRITING"+Z$
480 LET L$=N$(1):IF L$="*" THEN GOTO 5000
540 LET Z$=L$
600 PAUSE 0
1520 CLS:LET F2=PEEK 23658:POKE 23658,8
3080 INPUT #8;R$
5020 POKE 23658,F2:PRINT "END OF PROGRAM":STOP
```